AMSS DP Bellman Opt Conditions

Mathematica details

Demand System

We first summarize the consumer demand system

Tastes and Technology

In[4889]:=
```
fCMkt[zlab_] = MPLC zlab;
MPLC = MPLg = w = 1;
u[ℂ_, Thome_, 𝔾_] = uc[ℂ] + ug[𝔾 - gM] + ul[Thome];
BudCons = (ℂ + p b*) - (b + TR + wbar 𝕃);
Tbudg = (Thome + 𝕃) - Tendow;
Thome = Tendow - 𝕃;
NonNegVars = {ℂ};
```

The consumers objective is current utility plus the marginal utility value, λ^+, of holding b^* bonds at the beginning of the next period:

In[4896]:=
```
objCons = u[ℂ, Thome, 𝔾] + β λ⁺ b*
```

Out[4896]= $\beta \lambda_+ b^* + uc[\mathbb{C}] + ug[-g_M + \mathbb{G}] + ul[Tendow - \mathbb{L}]$

Fixed g

In[4897]:=
```
𝔾 = gM; ug[0] = 0;
```

In[4898]:=
```
λ⁺ = λ₊;
```

In[4899]:=
```
objCons
```

Out[4899]= $\beta \lambda_+ b^* + uc[\mathbb{C}] + ul[Tendow - \mathbb{L}]$

Consumer's Lagrangian

In[4900]:=
```
lagCons = objCons - λ* BudCons ;
```

Let's simplify the model by setting productivity to 1 and time endowment to 1.

In[4901]:=
```
wbar = 1 - τ; Tendow = 1;
Thome = Tendow - 𝕃;
```

The variables chosen by the consumer are $\{\mathbb{C}, \mathbb{L}, b^*\}$ and we compute the first-order conditions:

In[4903]:=
```
consVars =  {ℂ, 𝕃, b*};
```

In[4904]:=
```
focs = {D[lagCons, ℂ], D[lagCons, 𝕃], D[lagCons, b*]};
DmdCons = Join[focs, {BudCons}];
DmdCons // TableForm
```

Out[4906]//TableForm=

$$-\lambda^* + uc'[\mathbb{C}]$$
$$-(-1+\tau)\,\lambda^* - ul'[1-\mathbb{L}]$$
$$\beta\,\lambda_+ - p\,\lambda^*$$
$$-b - TR + \mathbb{C} - \mathbb{L}\,(1-\tau) + p\,b^*$$

DmdCons describes the consumer's decisions given the bond endowment, b, the tax policy, τ and TR, and the bond price, p.

Government Problem

Specify variables and bounds

Variables

varsToday are the choices that affect today's payoff to both government and agents.

In[4907]:=
```
varsToday = {ℂ, 𝕃, p, τ, TR};
```

varsTomorrow are today's choices for tomorrow's states

In[4908]:=
```
varsTomorrow = {b₊, λ₊};
```

In[4909]:=
```
vars = Join[varsToday, varsTomorrow]
```

Out[4909]=
```
{ℂ, 𝕃, p, τ, TR, b₊, λ₊}
```

Inequality bounds on variables

TR must be nonnegative

In[4910]:=
```
PolicyPos = {TR };
```

The set of all nonnegativity bounds is

In[4911]:=
```
GovPos = Join[PolicyPos]
```

Out[4911]=
```
{TR}
```

Objective

The government maximizes current utility plus discounted next period's value function.

```
In[4912]:= GovObj = u[ℂ, Thome, 𝔾] + β  V[b⁺, λ₊];
```

Constraints

Equality constraints

Government budget constraint

In[4913]:=
```
BudGov = - ( (w - wbar) (Tendow - Thome) + p b* - (b + 𝔾 + TR)) // Simplify;
```

The demand system equations (a.k.a. incentive compatibility conditions) are also equality constraints.

In[4914]:=
```
GovEqLHS = {DmdCons, BudGov} // Flatten;
GovEqLHS // TableForm
```

Out[4915]//TableForm=

$-\lambda^{*} + \mathrm{uc}'[\mathbb{C}]$

$-(-1+\tau)\,\lambda^{*} - \mathrm{ul}'[1-\mathbb{L}]$

$\beta\,\lambda_{+} - p\,\lambda^{*}$

$-b - TR + \mathbb{C} - \mathbb{L}\,(1-\tau) + p\,b^{*}$

$b + g_{M} + TR - \mathbb{L}\,\tau - p\,b^{*}$

Take the difference between the consumer and government budget constraint equations.

```
In[4916]:= GovEqLHS[[-1]] = GovEqLHS[[-1]] + GovEqLHS[[-2]] // Simplify;
            GovEqLHS // TableForm
```

Out[4917]//TableForm=

$$-\lambda^* + uc'[\mathbb{C}]$$
$$-(-1+\tau)\,\lambda^* - ul'[1-\mathbb{L}]$$
$$\beta\,\lambda_+ - p\,\lambda^*$$
$$-b - TR + \mathbb{C} - \mathbb{L}\,(1-\tau) + p\,b^*$$
$$g_M + \mathbb{C} - \mathbb{L}$$

We impose the promise-keeping constraint $\lambda = \lambda^*$ by replacing λ^* with λ.

```
In[4918]:= subsX = {b⁺ → b₊, λ* → λ, b* → b⁺};
```

```
In[4919]:= GovEqLHS = GovEqLHS //. subsX;
            GovEqLHS // TableForm
```

Out[4920]//TableForm=

$$-\lambda + uc'[\mathbb{C}]$$
$$-\lambda\,(-1+\tau) - ul'[1-\mathbb{L}]$$
$$-p\,\lambda + \beta\,\lambda_+$$
$$-b + b_+\,p - TR + \mathbb{C} - \mathbb{L}\,(1-\tau)$$
$$g_M + \mathbb{C} - \mathbb{L}$$

Inequality constraints

The only inequality constraints are the bounds on variables:

In[4921]:=
```
GovPos
```

Out[4921]=
```
{TR}
```

Dynamic Program

The dynamic programming problem is a difficult problem. At each state, one needs to solve a constrained optimization problem, possibly with nonconvex constraints.

In[4922]:=
```
GovObj
```

Out[4922]= $uc[\mathbb{C}] + ul[1 - \mathbb{L}] + \beta V[b^{+}, \lambda_{+}]$

In[4923]:=
```
Thread[GovEqLHS == 0] // TableForm
```

Out[4923]//TableForm=

$-\lambda + uc'[\mathbb{C}] == 0$

$-\lambda(-1 + \tau) - ul'[1 - \mathbb{L}] == 0$

$-p\lambda + \beta\lambda_{+} == 0$

$-b + b_{+}p - TR + \mathbb{C} - \mathbb{L}(1 - \tau) == 0$

$g_{M} + \mathbb{C} - \mathbb{L} == 0$

In[4924]:=
```
Thread[GovPos ≥ 0] // TableForm
```

Out[4924]//TableForm=

$TR \geq 0$

Calling solvers may be very costly

KKT system

Create the Lagrangian

Define shadow prices for equality constraints, and display a table that shows each shadow price and the corresponding equality constraint

In[4925]:=
```
ϕs = {ϕcIC, ϕlabIC, ϕeulIC, ϕcbud, ϕgbud};
{ϕs, GovEqLHS} // Transpose // TableForm
```

Out[4926]//TableForm=

ϕ_{cIC}	$-\lambda + uc'[\mathbb{C}]$
ϕ_{labIC}	$-\lambda\,(-1+\tau) - ul'[1-\mathbb{L}]$
ϕ_{eulIC}	$-p\,\lambda + \beta\,\lambda_+$
ϕ_{cbud}	$-b + b_+\,p - TR + \mathbb{C} - \mathbb{L}\,(1-\tau)$
ϕ_{gbud}	$g_M + \mathbb{C} - \mathbb{L}$

List inequality constraints and their multipliers.

In[4927]:=
```
GovIneq = Thread[GovPos ≥ 0];
μs = {μTR};
{μs, GovIneq} // Transpose // TableForm
```

Out[4929]//TableForm=

μ_{TR}	$TR \geq 0$

Create Lagrangian

In[4930]:=
```
lagGov = GovObj - ϕs.GovEqLHS + μs.GovPos /. subsX
```

Out[4930]=

$$TR\,\mu_{TR} - (-b + b_+\,p - TR + \mathbb{C} - \mathbb{L}\,(1-\tau))\,\phi_{cbud} - (-p\,\lambda + \beta\,\lambda_+)\,\phi_{eulIC} - (g_M + \mathbb{C} - \mathbb{L})\,\phi_{gbud} + uc[\mathbb{C}] + ul[1-\mathbb{L}] + \beta\,V[b_+, \lambda_+] - \phi_{cIC}\,(-\lambda + uc'[\mathbb{C}]) - \phi_{labIC}\,(-\lambda\,(-1+\tau) - ul'[1-\mathbb{L}])$$

Compute gradient of Lagrangian

List the choice variables of the government

In[4931]:=
```
govGradToday = Thread[D[lagGov, {varsToday}] == 0];
```

In[4932]:=
```
govGradTomorrow = Thread[D[lagGov, {varsTomorrow}] == 0];
```

Compute gradient of Lagangian w.r.t. the variables

In[4933]:=
```
govGrad = Join[govGradToday, govGradTomorrow];
govGrad // NTable
```

Out[4934]//TableForm=

	1
1	$-\phi_{cbud} - \phi_{gbud} + uc'[\mathbb{C}] - \phi_{cIC}\, uc''[\mathbb{C}] == 0$
2	$(1-\tau)\,\phi_{cbud} + \phi_{gbud} - ul'[1-\mathbb{L}] - \phi_{labIC}\, ul''[1-\mathbb{L}] == 0$
3	$-b_+\,\phi_{cbud} + \lambda\,\phi_{eulIC} == 0$
4	$-\mathbb{L}\,\phi_{cbud} + \lambda\,\phi_{labIC} == 0$
5	$\mu_{TR} + \phi_{cbud} == 0$
6	$-p\,\phi_{cbud} + \beta\, V^{(1,0)}[b_+, \lambda_+] == 0$
7	$-\beta\,\phi_{eulIC} + \beta\, V^{(0,1)}[b_+, \lambda_+] == 0$

Construct the KKT conditions

We next collect the inequality constraints and their complementarity conditions:

Recall the inequality constraints

In[4935]:=
```
GovIneq
```

Out[4935]=
```
{TR ≥ 0}
```

The multipliers must be nonnegative

In[4936]:=
```
mupos = Thread[μs ≥ 0]
```

Out[4936]= $\{\mu_{TR} \geq 0\}$

Construct the complementarity constraints:

In[4937]:=
```
govComps = Thread[μs GovPos == 0];
% // TableForm
```

Out[4938]//TableForm=

$TR\ \mu_{TR} == 0$

Combine them into single expressions

In[4939]:=
```
CompConditions = MapThread[And, {mupos, govComps, GovIneq}];
%% // NTable
```

Out[4940]//TableForm=

	1
1	$TR\ \mu_{TR} == 0$

Envelope theorem

In[4941]:=
```
lagGov
```

Out[4941]=

$$\text{TR}\,\mu_{\text{TR}} - (-b + b_{+}\,p - \text{TR} + \mathbb{C} - \mathbb{L}\,(1-\tau))\,\phi_{\text{cbud}} - (-p\,\lambda + \beta\,\lambda_{+})\,\phi_{\text{eulIC}} - (g_{M} + \mathbb{C} - \mathbb{L})\,\phi_{\text{gbud}} + \text{uc}[\mathbb{C}] + \text{ul}[1-\mathbb{L}] + \beta\,V[b_{+},\lambda_{+}] - \phi_{\text{cIC}}\,(-\lambda + \text{uc}'[\mathbb{C}]) - \phi_{\text{labIC}}\,(-\lambda\,(-1+\tau) - \text{ul}'[1-\mathbb{L}])$$

Use the envelope theorem to compute the gradients of the value function at the current state:

In[4942]:=
```
govEnvelope = {Vb[b, λ] == D[lagGov, b], Vλ[b, λ] == D[lagGov, λ]}
```

Out[4942]=

$$\{\text{Vb}[b,\lambda] == \phi_{\text{cbud}},\ \text{V}\lambda[b,\lambda] == \phi_{\text{cIC}} + p\,\phi_{\text{eulIC}} - (1-\tau)\,\phi_{\text{labIC}}\}$$

In[4943]:=
```
envgrads = {Vb[b, λ], Vλ[b, λ]};
```

In[4944]:=
```
envgradsol = Solve[govEnvelope, envgrads][[1]]
```

Out[4944]=

$$\{\text{Vb}[b,\lambda] \to \phi_{\text{cbud}},\ \text{V}\lambda[b,\lambda] \to \phi_{\text{cIC}} + p\,\phi_{\text{eulIC}} - \phi_{\text{labIC}} + \tau\,\phi_{\text{labIC}}\}$$

Collect all pieces into one system

Let’s rewrite the value function gradients to a more convenient notation

```
In[4945]:= notationSubs = {V^(1,0)[b+, zz_] → Vb[b+, zz], V^(0,1)[b+, zz_] → Vλ[b+, zz]};
```

Collect the equality constraints:

```
In[4946]:= GovEqLHS = Thread[GovEqLHS == 0];
```

Combine all constraints and complementarity conditions

In[4947]:=
```
TotSysGenAll = Join[
      GovEqLHS, govGradToday, govGradTomorrow, govEnvelope, CompConditions] /.
    notationSubs // Simplify;
% //
 NTable
```

Out[4948]//TableForm=

	1
1	$\lambda == \text{uc}'[\mathbb{C}]$
2	$\lambda == \lambda\,\tau + \text{ul}'[1 - \mathbb{L}]$
3	$p\,\lambda == \beta\,\lambda_{+}$
4	$b + TR + \mathbb{L} == b_{+}\,p + \mathbb{C} + \mathbb{L}\,\tau$
5	$g_{M} + \mathbb{C} == \mathbb{L}$
6	$\phi_{cbud} + \phi_{gbud} + \phi_{cIC}\,\text{uc}''[\mathbb{C}] == \text{uc}'[\mathbb{C}]$
7	$(-1 + \tau)\,\phi_{cbud} + \text{ul}'[1 - \mathbb{L}] + \phi_{labIC}\,\text{ul}''[1 - \mathbb{L}] == \phi_{gbud}$
8	$b_{+}\,\phi_{cbud} == \lambda\,\phi_{eulIC}$
9	$\mathbb{L}\,\phi_{cbud} == \lambda\,\phi_{labIC}$
10	$\mu_{TR} + \phi_{cbud} == 0$
11	$p\,\phi_{cbud} == \beta\,\text{Vb}[b_{+}, \lambda_{+}]$
12	$\beta\,\phi_{eulIC} == \beta\,\text{V}\lambda[b_{+}, \lambda_{+}]$
13	$\phi_{cbud} == \text{Vb}[b, \lambda]$
14	$\text{V}\lambda[b, \lambda] == \phi_{cIC} + p\,\phi_{eulIC} + (-1 + \tau)\,\phi_{labIC}$
15	$\mu_{TR} \geq 0\ \&\&\ TR\,\mu_{TR} == 0\ \&\&\ TR \geq 0$

In[4949]:=
```
TotSys = TotSysGen = TotSysGenAll;
```

Solutions for lambdas

Solve for Phis

We next solve for the shadow prices of the equality constraints.

In[4950]:= ϕs

Out[4950]= $\{\phi_{cIC}, \phi_{labIC}, \phi_{eulIC}, \phi_{cbud}, \phi_{gbud}\}$

These can be solved out from a subsystem of six equations of the TotSys system. These are the gradients of the optimization problem with respect to the contemporaneous variables. There are five ϕs, so we just need some five equations but they are defined by a set of six equations:

In[4951]:= TotSys[[6 ;; 11]] // TableForm

Out[4951]//TableForm=

$$\phi_{cbud} + \phi_{gbud} + \phi_{cIC}\, uc''[\mathbb{C}] == uc'[\mathbb{C}]$$
$$(-1+\tau)\,\phi_{cbud} + ul'[1-\mathbb{L}] + \phi_{labIC}\, ul''[1-\mathbb{L}] == \phi_{gbud}$$
$$b_+\,\phi_{cbud} == \lambda\,\phi_{eulIC}$$
$$\mathbb{L}\,\phi_{cbud} == \lambda\,\phi_{labIC}$$
$$\mu_{TR} + \phi_{cbud} == 0$$
$$p\,\phi_{cbud} == \beta\, Vb[b_+, \lambda_+]$$

We solve out for all shadow prices

In[4952]:=
```
phisols = Solve[TotSys[[6 ;; 10]], ϕs][[1]] // Apart;
% // NTable
```

Out[4953]//TableForm=

	1
1	$\phi_{\text{cIC}} \to \frac{\mu_{\text{TR}}\,\tau + \text{uc}'[\mathbb{C}] - \text{ul}'[1-\mathbb{L}]}{\text{uc}''[\mathbb{C}]} + \frac{\mathbb{L}\,\mu_{\text{TR}}\,\text{ul}''[1-\mathbb{L}]}{\lambda\,\text{uc}''[\mathbb{C}]}$
2	$\phi_{\text{labIC}} \to -\frac{\mathbb{L}\,\mu_{\text{TR}}}{\lambda}$
3	$\phi_{\text{eulIC}} \to -\frac{b_{+}\,\mu_{\text{TR}}}{\lambda}$
4	$\phi_{\text{cbud}} \to -\mu_{\text{TR}}$
5	$\phi_{\text{gbud}} \to \mu_{\text{TR}} - \mu_{\text{TR}}\,\tau + \text{ul}'[1-\mathbb{L}] - \frac{\mathbb{L}\,\mu_{\text{TR}}\,\text{ul}''[1-\mathbb{L}]}{\lambda}$

We eliminate the equality constraint shadow prices, and define CompSys the result

In[4954]:=
```
TotSysnolam = TotSys /. phisols // Simplify;
CompSys = TotSysnolam = % /. not;
% // NTable
```

Out[4956]//TableForm=

	1
1	$\lambda == uc'[\mathbb{C}]$
2	$\lambda == \lambda\,\tau + ul'[1-\mathbb{L}]$
3	$p\,\lambda == \beta\,\lambda_+$
4	$b + TR + \mathbb{L} == b_+\,p + \mathbb{C} + \mathbb{L}\,\tau$
5	$g_M + \mathbb{C} == \mathbb{L}$
6	$p\,\mu_{TR} + \beta\,Vb[b_+, \lambda_+] == 0$
7	$\beta\left(\frac{b_+\,\mu_{TR}}{\lambda} + V\lambda[b_+, \lambda_+]\right) == 0$
8	$\mu_{TR} + Vb[b, \lambda] == 0$
9	$V\lambda[b, \lambda] == \frac{\lambda\,uc'[\mathbb{C}] - \lambda\,ul'[1-\mathbb{L}] + \mu_{TR}\,(\lambda\,\tau + (-b_+\,p + \mathbb{L} - \mathbb{L}\,\tau)\,uc''[\mathbb{C}] + \mathbb{L}\,ul''[1-\mathbb{L}])}{\lambda\,uc''[\mathbb{C}]}$
10	$\mu_{TR} \geq 0\ \&\&\ TR\,\mu_{TR} == 0\ \&\&\ TR \geq 0$

In[4957]:=
```
DumpSave["AMSSsysFxgDet.mx", "Global`"]
```

Out[4957]= {Global`}